U.S. DEPARTMENT OF COMMERCE
Peter G. Peterson, Secretary

OFFICE OF TELECOMMUNICATIONS
Armig G. Kandoian, Director

INSTITUTE FOR TELECOMMUNICATION SCIENCES
Douglass D. Crombie, Acting Director

TELECOMMUNICATIONS
Research and Engineering Report 27

A METHOD OF NUMERICAL REPRESENTATION FOR THE AMPLITUDE-PROBABILITY DISTRIBUTION OF ATMOSPHERIC RADIO NOISE

HIROSHI AKIMA
INSTITUTE FOR TELECOMMUNICATION SCIENCES
Boulder, Colorado 80302

MARCH 1972 OT/TRER 27

For sale by the Superintendent of Documents, U. S. Government Printing Office, Washington, D. C. 20402
Price 35 cents

TABLE OF CONTENTS

A METHOD OF NUMERICAL REPRESENTATION FOR THE AMPLITUDE-PROBABILITY DISTRIBUTION OF ATMOSPHERIC RADIO NOISE

Hiroshi Akima

It is well known that the amplitude-probability distribution (APD) function of atmospheric radio noise can be closely represented by a curve that one can select from a family of curves by specifying the ratio of rms to average voltage, usually expressed in decibels and denoted by V_d. This report presents a method for numerically representing the APD function of atmospheric radio noise for a specified V_d value. It also presents two computer subprograms that implement the method.

Key Words: Amplitude-probability distribution (APD), atmospheric radio noise.

1. INTRODUCTION

To analyze the effects of radio noise on the performance of radio communication systems, we sometimes need knowledge of the amplitude-probability distribution (APD) of the noise, that is, the probability of a voltage level being exceeded by the envelope of the noise. For many applications, the APD function of atmospheric radio noise can be closely approximated by a curve that is selected from a family of curves (Spaulding et al., 1962; CCIR, 1964). One can make this selection by specifying the ratio of rms to average voltage, usually expressed in decibels and denoted by V_d.

Although CCIR Report 322 (CCIR, 1964) gives this family of curves, a method is needed for numerically representing the APD function to utilize those curves in computerized analysis of system

performances. In this study we develop such a method. The findings in previous studies (Crichlow et al., 1960; Spaulding et al., 1962) are fully utilized in this study. We also present two computer subprograms that implement this method.

2. DEVELOPMENT OF THE METHOD

2.1. Coordinate System

We denote the instantaneous envelope voltage of atmospheric noise, for which the APD function is desired, by v and the APD function by P(v). For the present, the unit of v is taken arbitrarily.

We use a coordinate system in which the abscissa and the ordinate are defined as

$$x = -20 \log(-\ln P), \tag{1}$$

$$y = 20 \log v, \tag{2}$$

respectively, where log is the common logarithm and ln the natural logarithm. With x and y, P and v are expressed by

$$P = \exp\{-\text{antilog}(-x/20)\}, \tag{3}$$

$$v = \text{antilog}(y/20), \tag{4}$$

where antilog is the inverse function of the common logarithm.

This coordinate system is shown in figure 1. The origin of the coordinate system corresponds to

$$P = 1/e = 0.367\ 879,$$

$$v = 1.$$

As shown in this figure, the Rayleigh distribution that is expressed by

$$P(v) = \exp(-v^2) \tag{5}$$



Figure 1. Coordinate system used.

plots as a straight line through the origin with a slope of -0.5 in this coordinate system. This coordinate system conforms to the custom followed by the previous studies (Crichlow et al., 1960; Spaulding et al., 1962) and the CCIR (1964).

2.2. Basic Assumption of Three-Section Curve

After Crichlow et al. (1960) and Spaulding et al. (1962), we assume that the APD function of atmospheric radio noise can be closely represented by a three-section curve in the coordinate system introduced in the previous section. This curve consists of two straight lines joined tangentially by a circular arc, as shown in figure 2.

For small y values the curve coincides with a straight line, L_1, having the same slope as the Rayleigh distribution, i.e., the slope of -0.5. For large y values the curve coincides with another straight line, L_2, having a steeper (negative) slope.

2.3. Geometry of the Three-Section Curve

We denote the points where the circular arc is tangent to straight lines L_1 and L_2 by (x_1, y_1) and (x_2, y_2), respectively, as shown in figure 2. We also denote the center of the circle by (x_c, y_c) and its radius by r. Then, the three-section curve can be expressed by

$$L_1: \quad y = m_1 x + b_1 \tag{6}$$

for $x \geq x_1$ and $y \leq y_1$;

$$C: \quad (x - x_c)^2 + (y - y_c)^2 = r^2 \tag{7}$$

for $x_1 \geq x \geq x_2$ and $y_1 \leq y \leq y_2$;



Figure 2. Three-section curve for the APD function.

$$L_2: \quad y = m_2 x + b_2 \tag{8}$$

for $x \leq x_2$ and $y \geq y_2$.

We denote the third straight line that is tangent to the arc at its midpoint by

$$L_3: \quad y = m_3 x + b_3 . \tag{9}$$

Since L_3 is parallel to the bisector of the angle between L_1 and L_2, m_3 is related to m_1 and m_2 by

$$m_3 = \tan\left(\frac{\tan^{-1} m_1 + \tan^{-1} m_2}{2}\right), \tag{10}$$

where a principal value in the range from $-\pi/2$ to $\pi/2$ is taken for the arctangent function. If we denote the direction tangent of a bisector of the angle between L_1 and L_3 by m_4, it is given by

$$m_4 = \tan\left(\frac{\tan^{-1} m_1 + \tan^{-1} m_3}{2}\right). \tag{11}$$

We further denote the point of intersection between L_1 and L_2 by (x_3, y_3), and between L_1 and L_3 by (x_4, y_4), as shown in figure 2. Then, these coordinates are given as

$$x_3 = \frac{b_2 - b_1}{m_1 - m_2}, \tag{12}$$

$$y_3 = \frac{m_1 b_2 - m_2 b_1}{m_1 - m_2}, \tag{13}$$

$$x_4 = \frac{b_3 - b_1}{m_1 - m_3}, \tag{14}$$

$$y_4 = \frac{m_1 b_3 - m_3 b_1}{m_1 - m_3} \quad . \tag{15}$$

With these values, the coordinates x_c and y_c are given by

$$x_c = \frac{m_3 (x_4 + m_4 y_4) - m_4 (x_3 + m_3 y_3)}{m_3 - m_4} \quad , \tag{16}$$

$$y_c = \frac{(x_3 + m_3 y_3) - (x_4 + m_4 y_4)}{m_3 - m_4} \quad . \tag{17}$$

The coordinates x_1, y_1, and y_2 are given by

$$x_1 = \frac{x_c + m_1 (y_c - b_1)}{1 + m_1^2} \quad , \tag{18}$$

$$y_1 = \frac{b_1 + m_1 x_c + m_1^2 y_c}{1 + m_1^2} \quad , \tag{19}$$

$$x_2 = \frac{x_c + m_2 (y_c - b_2)}{1 + m_2^2} \quad , \tag{20}$$

$$y_2 = \frac{b_2 + m_2 x_c + m_2^2 y_c}{1 + m_2^2} \quad . \tag{21}$$

Finally, the radius of the circular arc is expressed by

$$r = \sqrt{(x_c - x_1)^2 + (y_c - y_1)^2} \quad . \tag{22}$$

It follows from this geometry that, if five constants m_1, m_2, b_1, b_2, and b_3 are specified, all the necessary values for determining the curve by (6) to (8) can be uniquely determined by following steps from (10) to (22).

2.4. Additional Assumptions

Since straight line L_1 has the same slope as that for the Rayleigh distribution (Crichlow et al., 1960; Spaulding et al., 1962), we have

$$m_1 = -0.5 . \tag{23}$$

Spaulding et al. (1962) also showed that the third straight line L_3 can be expressed, as an empirical formula, by

$$y - y_3 = m_3 (x - x_3) + 1.5 (m_2 / m_1 - 1) .$$

From this, we have

$$b_3 = y_3 - m_3 x_3 + 1.5 (m_2 / m_1 - 1) . \tag{24}$$

Because of these additional conditions, we have now only three parameters, m_2, b_1, and b_2, that we must specify to construct the desired curve.

2.5. Moments of the APD

After Crichlow et al. (1960) and Spaulding et al. (1962), we introduce three moments that are used to specify the APD function of atmospheric radio noise. They are

rms voltage: $$v_{rms} = (\overline{v^2})^{1/2} , \tag{25}$$

average voltage: $$v_{ave} = \overline{v} , \tag{26}$$

logarithmic average voltage: $$v_{log} = \text{antilog}(\overline{\log v}) . \tag{27}$$

These moments can be expressed by

$$v_{rms} = \left\{ \int_{-\infty}^{\infty} [\text{antilog}(y/10)] (-dP/dy)\, dy \right\}^{1/2} , \tag{28}$$

$$v_{ave} = \int_{-\infty}^{\infty} [\text{antilog}(y/20)](-dP/dy)\,dy \ , \tag{29}$$

$$v_{log} = \text{antilog}\left[\int_{-\infty}^{\infty} (y/20)(-dP/dy)\,dy\right] \ , \tag{30}$$

where $(-dP/dy)$ is the probability density function. Since P is a function of x as given in (3), the probability density function can be expressed by

$$(-dP/dy) = (dP/dx)(-dx/dy)$$

$$= [\exp\{-\text{antilog}(-x/20)\}]\,[\text{antilog}(-x/20)]\,(1/20M)(-dx/dy) \ , \tag{31}$$

where M is a constant defined by

$$M = \log_{10} e = 0.434\,294 \ . \tag{32}$$

Therefore, if $x = x(y)$ is given as a function of y in our coordinate system, we can calculate the three moments by numerical integration from (28) through (30).

We use v_{rms} to represent the intensity of atmospheric radio noise, and the ratios of v_{rms} to the other two moments to represent the statistical characteristics of the atmospheric radio noise or the shape of its APD function. These ratios are expressed in decibels, denoted by V_d and L_d, and defined as

$$V_d = 20 \log (v_{rms}/v_{ave}) \ , \tag{33}$$

$$L_d = 20 \log (v_{rms}/v_{log}) \ . \tag{34}$$

If we move the curve vertically without deforming it, the magnitudes of v_{rms}, v_{ave}, and v_{log} vary by a common factor and neither V_d nor L_d will vary, because the ordinate of our coordinate system is envelope voltage expressed in decibels. Since moving the curve vertically without deforming it corresponds to changing b_1 and b_2 without

changing m_2 and $(b_1 - b_2)$, we can see that both V_d and L_d are functions of m_2 and $(b_1 - b_2)$ and are independent of b_1 or b_2 itself.

Setting $b_1 = 0$ momentarily, we determined the APD function for each set of m_2 and $(b_1 - b_2)$ values from (10) through (24), calculated v_{rms}, v_{ave}, and v_{log} by numerical integration from (28) through (30), and calculated V_d and L_d from (33) and (34), respectively. The results are shown in figures 3 and 4 as their respective contour maps. Note that, for the Rayleigh distribution, $V_d = 1.05$ dB and $L_d = 2.51$ dB. We can use these figures to determine the set of necessary values of m_2 and $(b_1 - b_2)$ for a set of specified values of V_d and L_d by overlapping two respective contours on a sheet and determining the point of intersection of the two contours.

2.6. Correlation of the Two Moments

Spaulding et al. (1962) showed that observed values of V_d and L_d are very highly correlated. This fact enables us to simplify the procedure of determining the APD function and to determine both m_2 and $(b_1 - b_2)$ by specifying only one parameter.

From figure 1 of their paper, we determined the average relation between observed V_d and L_d as

$$L_d = 1.697\, V_d + 0.7265 \ . \tag{35}$$

Using now the calculated values of V_d and L_d given by (28) - (34), we calculated

$$\Delta = L_d - (1.697\, V_d + 0.7265) \tag{36}$$

as a function of m_2 and $(b_1 - b_2)$. The result is shown in figure 5 as a contour map. This figure indicates that the contour for $\Delta = 0$ dB behaves in a very strange manner. But, if we relax (35) and allow a



Figure 3. Contour map of V_d.



Figure 4. Contour map of L_d.



Figure 5. Contour map of Δ.

deviation from it by a small amount that does not exceed 0.3 dB, we can draw a smooth and monotonic increasing curve, as shown in this figure. We adopt this curve as a basic relation between m_2 and $(b_1 - b_2)$.

We could draw another smooth and monotonic increasing curve that has a steeper slope than the adopted curve. For example, the contour for a small value of V_d in figure 3 gives a small deviation from (35). Such a curve, however, does not lead to actually observed APD function of atmospheric radio noise (Spaulding, private communication). Moreover, it cannot cover large values of V_d, say 5 dB or greater. Therefore, we disregard such a curve as an extraneous solution.

2.7. Determination of Necessary Parameters

Once the numerical relation between m_2 and $(b_1 - b_2)$ is determined, we can establish a method that allows us to construct a curve for the APD function by specifying only one parameter, V_d. All necessary values of parameters are determined in the following manner.

We first assign a value to m_2 and next read the value of $(b_1 - b_2)$ for this assigned m_2 value from figure 5. Setting $b_1 = 0$ momentarily, we calculate the necessary b and m constants, determine the APD function from (10) through (24), calculate v_{rms} and v_{ave} by numerical integration from (28) and (29), and calculate V_d from (33). Although b_1 could take any value, it is convenient to assign a value of $-20 \log v_{rms}$ to b_1, where v_{rms} is the value calculated with $b_1 = 0$. By this option, we can make the value of v_{rms} calculated with the new b_1 value to be unity (or 0 dB) and let the variable v denote the envelope voltage normalized with its rms value as a unit.

The values of b_1, b_2, and V_d thus calculated for several m_2 values are shown in table 1. Since b_1, b_2, and V_d are smooth and monotonic functions of m_2, we can interpolate values of m_2, b_1 and

Table 1. Computed Dependence of b_1, b_2, and V_d upon m_2

m_2	b_1	b_2	V_d
-0.5	0.0000	0.0000	1.0491
-0.6	-0.4329	-0.7529	1.1779
-0.7	-0.8909	-1.5309	1.3215
-0.8	-1.3751	-2.3351	1.4803
-0.9	-1.8867	-3.1667	1.6549
-1.0	-2.4269	-4.0269	1.8466
-1.2	-3.5983	-5.8383	2.2831
-1.4	-4.8927	-7.7827	2.7973
-1.6	-6.3195	-9.8695	3.3941
-1.8	-7.8868	-12.1068	4.0796
-2.0	-9.5991	-14.4991	4.8567
-2.2	-11.4495	-17.0495	5.7218
-2.4	-13.4448	-19.7548	6.6744
-2.6	-15.5800	-22.6100	7.7069
-2.8	-17.8472	-25.6072	8.8107
-3.0	-20.2380	-28.7380	9.9740
-3.5	-26.3694	-37.0919	12.9794
-4.0	-32.6321	-46.0824	16.0528
-5.0	-44.9001	-65.6023	22.1551
-6.0	-57.0708	-86.8042	28.2294
-7.0	-69.2146	-109.4042	32.2720
-8.0	-81.3777	-133.2062	40.2839
-9.0	-93.6426	-158.0634	46.2711
-10.0	-105.8298	-183.8612	52.2264

b_2 for a given V_d value. Therefore, we can uniquely construct a curve for the APD function for a given V_d value.

2.8. Comparison With the CCIR Curves

CCIR Report 322 (CCIR, 1964) gives the APD curves parametric in V_d for V_d = 2 to 30 dB at a 2-dB interval. It also gives a curve that corresponds to V_d = 1.05 dB, i.e., the curve for the Rayleigh distribution. We calculated the APD function, P, for the same values of V_d by the method described in this report and compared the result with the CCIR curves.

For $V_d \geq 8$ dB, the APD curve generated by this method differs from the respective CCIR curve by less than 1 dB. For V_d = 2 to 6 dB, the APD curve by this method falls within 1 dB of the respective CCIR curve for $P \geq 10^{-3}$, but the former is lower than the latter by more than 1 dB for $P < 10^{-3}$. In other words, the slope of the curve by this method is more gentle than that of the CCIR curve in the range of small P values.

The normalized envelope voltage, v, at $P = 10^{-6}$ calculated by this method is smaller than the CCIR value by as much as 3 dB for V_d = 2 to 6 dB. However, the v values at $P = 10^{-6}$ for equally spaced V_d values calculated by this method are more regularly spaced than the CCIR values. This seems to be a rather desirable feature for the purpose of presenting a family of idealized APD curves of atmospheric radio noise.

3. CONCLUSIONS

We have described a method for numerically representing the APD function of atmospheric radio noise for a specified value of V_d, the ratio of rms to average of envelope voltage of the noise. This method is mostly based on previous studies by Crichlow et al. (1960) and Spaulding et al. (1962); the only exception is that we relax the relation between V_d and L_d (35) by a small amount to obtain a smooth curve for the relation between some necessary parameters.

This method was implemented in computer subprograms. These subprograms were successfully used by the author (Akima et al., 1969; Akima, 1970) and proved to be useful. They are described in the appendix. A similar mathematical representation of the APD function for atmospheric radio noise has been used by various authors in system performance studies. See, for example, Conda (1965), Halton and Spaulding (1966), and Spaulding (1966).

4. ACKNOWLEDGMENTS

The author expresses his deep appreciation to Edwin L. Crow, Robert W. Hubbard, and Arthur D. Spaulding of the Institute for Telecommunication Sciences for their constructive criticisms on this report.

5. REFERENCES

Akima, Hiroshi (1970), Modulation studies for IGOSS, ESSA Tech. Rept. ERL 172 - ITS 110 (U.S. Government Printing Office, Washington, D.C. 20402).

Akima, Hiroshi, Gene G. Ax, and Wesley M. Beery (1969), Required signal-to-noise ratios for HF communication systems, ESSA Tech. Rept. ERL 131 - ITS 92 (U.S. Government Printing Office, Washington , D.C. 20402).

CCIR (International Radio Consultative Committee) (1964), World distribution and characteristics of atmospheric radio noise, CCIR Report 322 (International Telecommunication Union, Geneva, Switzerland).

Conda, Alyce M. (1965), The effect of atmospheric noise on the probability of error for an NCFSK system, IEEE Trans. Commun. Technol. COM - 13, 280 - 284.

Crichlow, W. Q., C. J. Roubique, A. D. Spaulding, and W. M. Beery (1960), Determination of the amplitude-probability distribution of atmospheric radio noise from statistical moments, J. Res. NBS 64D, 49 - 56.

Halton, J. H., and A. D. Spaulding (1966), Error rates in differentially coherent phase systems in non-Gaussian noise, IEEE Trans. Commun. Technol. COM - 14, 594 - 601.

Spaulding, Arthur D. (1966), The characteristics of atmospheric radio noise and its effects on digital communication systems, 1966 IEEE International Communications Conference, Paper No. CP - 1126.

Spaulding, A. D., C. J. Roubique, and W. Q. Crichlow (1962), Conversion of the amplitude-probability distribution function for atmospheric radio noise from one bandwidth to another, J. Res. NBS 66D, 713 - 720.

APPENDIX

Computer Subprograms

User information and Fortran listings are given on two function subprograms, APDAN and RANAN. Each subprogram implements the method for numerically representing the amplitude-probability distribution (APD) function of atmospheric radio noise, described in the text. These subprograms are written in the CDC-3800 Fortran. (For detail of the CDC-3800 Fortran, see 3400/3600/3800 Computer Systems Fortran Reference Manual, Pub. No. 60132900, and 3600/3800 Computer Systems Library Functions, Pub. No. 60056400 B, Control Data Corporation, Palo Alto, Calif.)

The APDAN Function. This function gives a value of the APD function or its density function of atmospheric radio noise for a specified V_d value and for a specified envelope voltage relative to its rms value.

This function has the form

APDAN(VD, K, DB)

where

VD = CCIR-NBS atmospheric noise parameter
= ratio of rms to average of envelope voltage expressed in decibels,

K = 1 for the APD function
= 2 for the probability density function,

DB = envelope voltage, in decibels relative to its rms value, for which the APD or probability density function is to be computed.

This function occupies 478 locations on the CDC-3800 computer. Computation time required for this function on the same computer is

approximately equal to 1500 microseconds when this function is called with a VD value that is different from the one in the previous call, and to 500 microseconds when called with the same VD value as in the previous call.

The RANAN Function. Successive calls of this function generate a series of random numbers that follow the APD function of atmospheric radio noise having a specified V_d value and its rms value of unity.

This function has the form

RANAN(VD)

where

VD = CCIR-NBS atmospheric noise parameter
= ratio of rms to average of envelope voltage expressed in decibels.

This function occupies 406 locations on the CDC-3800 computer. Computation time required for this function on the same computer is approximately equal to 1600 microseconds when this function is called with a different VD value from the one in the previous call, and to 500 microseconds when called with the same VD value.

```
      FUNCTION  APDAN(VD,K,DB)                                                AA  001
C     AMPLITUDE-PROBABILITY DISTRIBUTION OF ATMOSPHERIC NOISE                 AA  002
                                                                              AA  003
C     VD = CCIR-NBS NOISE PARAMETER                                           AA  004
C        = RATIO OF RMS TO AVERAGE OF NOISE ENVELOPE IN DB                    AA  005
C     K  = 1  FOR AMPLITUDE-PROBABILITY DISTRIBUTION FUNCTION                 AA  006
C        = 2  FOR PROBABILITY DENSITY FUNCTION                                AA  007
C     DB = LEVEL RELETIVE TO RMS IN DB                                        AA  008
                                                                              AA  009
                                                                              AA  010
C     DECLARATION STATEMENTS                                                  AA  011
                                                                              AA  012
      DIMENSION   VVD(24),BB1(24),BB2(24),MM2(24)                             AA  013
      REAL        MM2,M1,M2,M3,M4,M1SQP1,M2SQP1                               AA  014
      DATA  (VVD=                                                             AA  015
     1      1.0491,   1.1779,   1.3215,   1.4803,   1.6549,   1.8466,        AA  016
     2      2.2831,   2.7973,   3.3941,   4.0796,   4.8567,   5.7218,        AA  017
     3      6.6744,   7.7069,   8.8107,   9.9740,  12.9794,  16.0528,        AA  018
     4     22.1551,  28.2294,  34.2720,  40.2839,  46.2711,  52.2264)        AA  019
      DATA  (BB1=                                                             AA  020
     1      0.0000,  -0.4329,  -0.8909,  -1.3751,  -1.8867,  -2.4269,        AA  021
     2     -3.5983,  -4.8927,  -6.3195,  -7.8868,  -9.5991, -11.4495,        AA  022
     3    -13.4448, -15.5800, -17.8472, -20.2380, -26.3694, -32.6321,        AA  023
     4    -44.9001, -57.0708, -69.2146, -81.3777, -93.6426,-105.8298)        AA  024
      DATA  (BB2=                                                             AA  025
     1      0.0000,  -0.7529,  -1.5309,  -2.3351,  -3.1667,  -4.0269,        AA  026
     2     -5.8383,  -7.7827,  -9.8695, -12.1068, -14.4991, -17.0495,        AA  027
     3    -19.7548, -22.6100, -25.6072, -28.7380, -37.0919, -46.0824,        AA  028
     4    -65.6023, -86.8042,-109.4042,-133.2062,-158.0634,-183.8612)        AA  029
      DATA  (MM2=                                                             AA  030
     1     -0.5000,  -0.6000,  -0.7000,  -0.8000,  -0.9000,  -1.0000,        AA  031
     2     -1.2000,  -1.4000,  -1.6000,  -1.8000,  -2.0000,  -2.2000,        AA  032
     3     -2.4000,  -2.6000,  -2.8000,  -3.0000,  -3.5000,  -4.0000,        AA  033
     4     -5.0000,  -6.0000,  -7.0000,  -8.0000,  -9.0000, -10.0000)        AA  034
      DATA  (M1=-0.5),(M1SQP1=1.25),                                          AA  035
     1      (C1=0.2302585093),(C2=0.1151292546),                             AA  036
     2      (VDPV=0.0)                                                       AA  037
      EQUIVALENCE (I,DM,L),(IMN,V1,B1),(IMX,V2,B2),(V3,M2),(V4,SF),           AA  038
     1            (V43,XC),(V42,YC),(V41,DBMN),(V32,DBMX),                  AA  039
     2            (V31,RSQ),(V21,SM,T),                                     AA  040
     3            (A1,X3,DX),(A2,Y3,DY),(A3,X4,PP),(A4,Y4)                  AA  041
                                                                              AA  042
C     CHECK IF INPUT PARAMETERS ARE IN ERROR                                  AA  043
                                                                              AA  044
   10 VD0=VD              $    K0=K              $    DB0=DB                  AA  045
      IF(VD0.LT.1.049)         GO TO 90                                       AA  046
      IF(K0.LT.1.OR.K0.GT.2)   GO TO 91                                       AA  047
                                                                              AA  048
C     CHECK IF VD IS THE SAME AS IN THE PREVIOUS CALL                         AA  049
                                                                              AA  050
   20 IF(VD0.GE.1.05)          GO TO 22                                       AA  051
   21 L=1                                                                     AA  052
      PP=EXP(C1*DB0)       $    GO TO 65                                     AA  053
   22 IF(VD0.EQ.VDPV)          GO TO 60                                       AA  054
                                                                              AA  055
C     LOCATE VD0                                                              AA  056
                                                                              AA  057
   30 VDPV=VD0                                                                AA  058
      IF(VD0.LT.VVD(3))        GO TO 35                                       AA  059
      IF(VD0.GE.VVD(22))       GO TO 36                                       AA  060
      IMN=4                $    IMX=22                                       AA  061
   31 I=(IMN+IMX)/2                                                           AA  062
      IF(VD0.GE.VVD(I))        GO TO 33                                       AA  063
   32 IMX=I                $    GO TO 34                                     AA  064
   33 IMN=I+1                                                                 AA  065
   34 IF(IMX.GT.IMN)           GO TO 31                                       AA  066
      I=IMX                $    GO TO 40                                     AA  067
```

```
   35 I=3                  $    GO TO 40                                    AA  068
   36 I=23                                                                  AA  069
                                                                            AA  070
C     INTERPOLATION OF B1, B2, AND M2                                       AA  071
                                                                            AA  072
   40 V1=VVD(I-2)-VD0      $    V2=VVD(I-1)-VD0                             AA  073
      V3=VVD(I)  -VD0      $    V4=VVD(I+1)-VD0                             AA  074
      V43=V4-V3            $    V42=V4-V2           $    V41=V4-V1         AA  075
      V32=V3-V2            $    V31=V3-V1            $    V21=V2-V1         AA  076
      A1=V41*V31*V21       $    A1= V4*V3*V2/A1                            AA  077
      A2=V42*V32*V21       $    A2=-V4*V3*V1/A2                            AA  078
      A3=V43*V32*V31       $    A3= V4*V2*V1/A3                            AA  079
      A4=V43*V42*V41       $    A4=-V3*V2*V1/A4                            AA  080
      B1=A1*BB1(I-2)+A2*BB1(I-1)+A3*BB1(I)+A4*BB1(I+1)                      AA  081
      B2=A1*BB2(I-2)+A2*BB2(I-1)+A3*BB2(I)+A4*BB2(I+1)                      AA  082
      M2=A1*MM2(I-2)+A2*MM2(I-1)+A3*MM2(I)+A4*MM2(I+1)                      AA  083
                                                                            AA  084
C     GEOMETRY                                                              AA  085
                                                                            AA  086
   50 SF=M2/M1             $    M2SQP1=M2*M2+1.0                           AA  087
      DM=M1-M2             $    SM=M1+M2                                   AA  088
      X3=(B2-B1)/DM        $    Y3=(M1*B2-M2*B1)/DM                         AA  089
      T=(1.0-M1*M2)/SM     $    M3=-T-SQRT(T*T+1.0)                          AA  090
      B3=Y3-M3*X3+1.5*(SF-1.0)                                              AA  091
      DM=M1-M3             $    SM=M1+M3                                   AA  092
      X4=(B3-B1)/DM        $    Y4=(M1*B3-M3*B1)/DM                         AA  093
      T=(1.0-M1*M3)/SM     $    M4=-T-SQRT(T*T+1.0)                          AA  094
      DM=M3-M4                                                              AA  095
      X3=X3+M3*Y3          $    X4=X4+M4*Y4                                 AA  096
      XC=(M3*X4-M4*X3)/DM  $    YC=(X3-X4)/DM                                AA  097
      DBMN=(B1+M1*(XC+M1*YC))/M1SQP1                                        AA  098
      DBMX=(B2+M2*(XC+M2*YC))/M2SQP1                                        AA  099
      RSQ=(YC-DBMN)**2*M1SQP1                                               AA  100
                                                                            AA  101
C     COMPUTATION OF THE FUNCTION                                           AA  102
                                                                            AA  103
   60 IF(DB0.GE.DBMX)         GO TO 63                                      AA  104
      IF(DB0.GT.DBMN)         GO TO 62                                      AA  105
   61 L=1                                                                   AA  106
      PP=EXP(C1*(DB0-B1))  $  GO TO 65                                      AA  107
   62 L=2                                                                   AA  108
      DY=YC-DB0            $  DX=SQRT(RSQ-DY*DY)                           AA  109
      PP=EXP(C2*(DX-XC))   $  GO TO 65                                      AA  110
   63 L=3                                                                   AA  111
      PP=EXP(C1*(DB0-B2)/SF)                                                AA  112
   65 APDAN=EXP(-PP)                                                        AA  113
      IF(K0.EQ.1)             RETURN                                        AA  114
      GO TO (69,67,68), L                                                   AA  115
   67 APDAN=0.5*DY*APDAN/DX   $    GO TO 69                                 AA  116
   68 APDAN=APDAN/SF                                                        AA  117
   69 APDAN=APDAN*PP*C1    $  RETURN                                        AA  118
                                                                            AA  119
C     ERROR EXIT                                                            AA  120
                                                                            AA  121
   90 PRINT 2090           $  GO TO 95                                      AA  122
   91 PRINT 2091                                                            AA  123
   95 PRINT 2095, VD0,K0,DB0                                                AA  124
      RETURN                                                                AA  125
                                                                            AA  126
C     FORMAT STATEMENTS                                                     AA  127
                                                                            AA  128
 2090 FORMAT(1X/21H  ***   VD TOO SMALL./)                                  AA  129
 2091 FORMAT(1X/25H  ***   IMPROPER K VALUE./)                              AA  130
 2095 FORMAT(7H   VD =,E12.3,8X,3HK =,I3,8X,4HDB =, E12.3/                  AA  131
     1       35H ERROR DETECTED IN ROUTINE    APDAN)                       AA  132
      END                                                                   AA  133-
```

```
      FUNCTION  RANAN(VD)                                                    RA  001
C     RANDOM NUMBER THAT FOLLOWS THE APD OF ATMOSPHERIC NOISE                RA  002
C     THIS FUNCTION GIVES A LEVEL IN DB RELATIVE TO RMS                      RA  003
                                                                             RA  004
C     VD = CCIR-NBS NOISE PARAMETER                                          RA  005
C        = RATIO OF RMS TO AVERAGE OF NOISE ENVELOPE IN DB                   RA  006
                                                                             RA  007
                                                                             RA  008
C     DECLARATION STATEMENTS                                                 RA  009
                                                                             RA  010
      DIMENSION   VVD(24),BB1(24),BB2(24),MM2(24)                            RA  011
      REAL        MM2,M1,M2,M3,M4,M1SQP1,M2SQP1                              RA  012
      DATA  (VVD=                                                            RA  013
     1      1.0491,   1.1779,   1.3215,   1.4803,   1.6549,   1.8466,       RA  014
     2      2.2831,   2.7973,   3.3941,   4.0796,   4.8567,   5.7218,       RA  015
     3      6.6744,   7.7069,   8.8107,   9.9740,  12.9794,  16.0528,       RA  016
     4     22.1551,  28.2294,  34.2720,  40.2839,  46.2711,  52.2264)       RA  017
      DATA  (BB1=                                                            RA  018
     1      0.0000,  -0.4329,  -0.8909,  -1.3751,  -1.8867,  -2.4269,       RA  019
     2     -3.5983,  -4.8927,  -6.3195,  -7.8868,  -9.5991, -11.4495,       RA  020
     3    -13.4448, -15.5800, -17.8472, -20.2380, -26.3694, -32.6321,       RA  021
     4    -44.9001, -57.0708, -69.2146, -81.3777, -93.6426,-105.8298)       RA  022
      DATA  (BB2=                                                            RA  023
     1      0.0000,  -0.7529,  -1.5309,  -2.3351,  -3.1667,  -4.0269,       RA  024
     2     -5.8383,  -7.7827,  -9.8695, -12.1068, -14.4991, -17.0495,       RA  025
     3    -19.7548, -22.6100, -25.6072, -28.7380, -37.0919, -46.0824,       RA  026
     4    -65.6023, -86.8042,-109.4042,-133.2062,-158.0634,-183.8612)       RA  027
      DATA  (MM2=                                                            RA  028
     1     -0.5000,  -0.6000,  -0.7000,  -0.8000,  -0.9000,  -1.0000,       RA  029
     2     -1.2000,  -1.4000,  -1.6000,  -1.8000,  -2.0000,  -2.2000,       RA  030
     3     -2.4000,  -2.6000,  -2.8000,  -3.0000,  -3.5000,  -4.0000,       RA  031
     4     -5.0000,  -6.0000,  -7.0000,  -8.0000,  -9.0000, -10.0000)       RA  032
      DATA  (M1=-0.5),(M1SQP1=1.25),(C0=8.685889638),(VDPV=0.0)              RA  033
      EQUIVALENCE (PROB,X),(I,DM),                                           RA  034
     1            (IMN,V1,B1),(IMX,V2,B2),(V3,M2),(V4,SF),                  RA  035
     2            (V43,XC),(V42,YC),(V41,XMX),(V32,XMN),                    RA  036
     3            (V31,RSQ),(V21,SM,T),                                     RA  037
     3            (A1,X3),(A2,Y3),(A3,X4),(A4,Y4)                           RA  038
                                                                             RA  039
C     CHECK IF INPUT PARAMETER IS IN ERROR                                   RA  040
                                                                             RA  041
   10 VD0=VD                                                                 RA  042
      IF(VD0.LT.1.049)          GO TO 90                                     RA  043
                                                                             RA  044
C     GENERATION OF A RANDOM X VARIABLE                                      RA  045
                                                                             RA  046
   15 PROB=RANF(-1)                                                          RA  047
      IF(PROB.LE.0.0)           GO TO 15                                     RA  048
      IF(PROB.GE.1.0)           GO TO 15                                     RA  049
      X=-C0*ALOG(-ALOG(PROB))                                                RA  050
                                                                             RA  051
C     CHECK IF VD IS THE SAME AS IN THE PREVIOUS CALL                        RA  052
                                                                             RA  053
   20 IF(VD0.GE.1.05)           GO TO 22                                     RA  054
   21 RANAN=M1*X             $    RETURN                                     RA  055
   22 IF(VD0.EQ.VDPV)           GO TO 60                                     RA  056
                                                                             RA  057
C     LOCATE VD0                                                             RA  058
                                                                             RA  059
   30 VDPV=VD0                                                               RA  060
      IF(VD0.LT.VVD(3))         GO TO 35                                     RA  061
      IF(VD0.GE.VVD(22))        GO TO 36                                     RA  062
      IMN=4                  $    IMX=22                                     RA  063
   31 I=(IMN+IMX)/2                                                          RA  064
      IF(VD0.GE.VVD(I))         GO TO 33                                     RA  065
   32 IMX=I                  $    GO TO 34                                   RA  066
   33 IMN=I+1                                                                RA  067
```

```
   34 IF(IMX.GT.IMN)              GO TO 31                                          RA  068
      I=IMX                  $    GO TO 40                                          RA  069
   35 I=3                    $    GO TO 40                                          RA  070
   36 I=23                                                                         RA  071
                                                                                   RA  072
C     INTERPOLATION OF B1, B2, AND M2                                             RA  073
                                                                                   RA  074
   40 V1=VVD(I-2)-VD0        $    V2=VVD(I-1)-VD0                                  RA  075
      V3=VVD(I)  -VD0        $    V4=VVD(I+1)-VD0                                  RA  076
      V43=V4-V3              $    V42=V4-V2              $    V41=V4-V1           RA  077
      V32=V3-V2              $    V31=V3-V1               $    V21=V2-V1           RA  078
      A1=V41*V31*V21         $    A1= V4*V3*V2/A1                                  RA  079
      A2=V42*V32*V21         $    A2=-V4*V3*V1/A2                                  RA  080
      A3=V43*V32*V31         $    A3= V4*V2*V1/A3                                  RA  081
      A4=V43*V42*V41         $    A4=-V3*V2*V1/A4                                  RA  082
      B1=A1*BB1(I-2)+A2*BB1(I-1)+A3*BB1(I)+A4*BB1(I+1)                             RA  083
      B2=A1*BB2(I-2)+A2*BB2(I-1)+A3*BB2(I)+A4*BB2(I+1)                             RA  084
      M2=A1*MM2(I-2)+A2*MM2(I-1)+A3*MM2(I)+A4*MM2(I+1)                             RA  085
                                                                                   RA  086
C     GEOMETRY                                                                    RA  087
                                                                                   RA  088
   50 SF=M2/M1               $    M2SQP1=M2*M2+1.0                                RA  089
      DM=M1-M2               $    SM=M1+M2                                         RA  090
      X3=(B2-B1)/DM          $    Y3=(M1*B2-M2*B1)/DM                               RA  091
      T=(1.0-M1*M2)/SM       $    M3=-T-SQRT(T*T+1.0)                               RA  092
      B3=Y3-M3*X3+1.5*(SF-1.0)                                                     RA  093
      DM=M1-M3               $    SM=M1+M3                                         RA  094
      X4=(B3-B1)/DM          $    Y4=(M1*B3-M3*B1)/DM                               RA  095
      T=(1.0-M1*M3)/SM       $    M4=-T-SQRT(T*T+1.0)                               RA  096
      DM=M3-M4                                                                     RA  097
      X3=X3+M3*Y3            $    X4=X4+M4*Y4                                      RA  098
      XC=(M3*X4-M4*X3)/DM    $    YC=(X3-X4)/DM                                     RA  099
      XMX=(XC+M1*(YC-B1))/M1SQP1                                                   RA  100
      XMN=(XC+M2*(YC-B2))/M2SQP1                                                   RA  101
      RSQ=5.0*(XMX-XC)**2                                                          RA  102
                                                                                   RA  103
C     COMPUTATION OF THE FUNCTION                                                 RA  104
                                                                                   RA  105
   60 IF(X.LE.XMN)                GO TO 63                                        RA  106
      IF(X.LT.XMX)                GO TO 62                                        RA  107
   61 RANAN=M1*X+B1          $    RETURN                                          RA  108
   62 RANAN=YC-SQRT(RSQ-(X-XC)**2)  $    RETURN                                  RA  109
   63 RANAN=M2*X+B2          $    RETURN                                          RA  110
                                                                                   RA  111
C     ERROR EXIT                                                                  RA  112
                                                                                   RA  113
   90 PRINT 2090, VD0        $    RETURN                                          RA  114
                                                                                   RA  115
C     FORMAT STATEMENT                                                            RA  116
                                                                                   RA  117
 2090 FORMAT(1X/21H  ***   VD TOO SMALL.//7H   VD =,E12.3/                        RA  118
     1       35H ERROR DETECTED IN ROUTINE    RANAN)                             RA  119
      END                                                                          RA  120-
```